Heat Index
Heat Index

This application note defines the term heat index, provides an equation for estimating it, and includes a datalogger program.

Definition

High humidity combined with hot temperatures reduce the body's ability to cool itself increasing the risk of heat exhaustion, heat stroke, and other heat related health problems. The Heat Index, also referred to as apparent temperature, is an estimate of the temperature (in °F) that would similarly affect the body at normal humidity (about 20 percent). For example, if the actual temperature is 100°F with 40 percent relative humidity, the heat index is 110°F meaning the apparent temperature feels like 110°F to the body.

Heat Index Equation¹

The following equation assumes the temperature is measured in the shade and at sea level, and the wind speed is 6 mph. Exposure to full sunshine can increase the heat index 5° to 15°F. Wind speed has a minimal effect on the Heat Index.

\[
HI = -42.379 + 2.0490(T_f) + 10.143(RH) - 0.22476(T_f)(RH) - (6.8378 \times 10^{-3})(T_f^2) - (5.4817 \times 10^{-2})RH^2 + (1.2287 \times 10^{-3})(T_f^2)RH + (8.5282 \times 10^{-4})(RH^2)(T_f) - (1.99 \times 10^{-6})(T_f^2)(RH^2)
\]

Where: HI = heat index in degrees Fahrenheit

\[T_f = \text{temperature in degrees Fahrenheit}\]

\[RH = \text{relative humidity in percent form (e.g., for an RH value of 65% enter 65 in the formula not .65)}\]

The equation is only valid for temperatures greater than or equal to 80°F and RH levels greater than or equal to 40%.
Program Example

The datalogger can be programmed to automatically calculate heat index. Below is the portion of a CR10X program that uses an Edlog expression to calculate heat index. Temperature and relative humidity are measured with a HMP45C Temperature and RH sensor.

{CR10X}

*Table 1 Program
  01:  60 Execution Interval (seconds)

;Turn the HMP45C on

1:  Do (P86)
    1:  41 Set Port 1 High

;Pause 150 milliseconds before making measurements
;to allow the probe to stabilize on true readings

2:  Excitation with Delay (P22)
    1:  1 Ex Channel
    2:  0 Delay W/Ex (units = 0.01 sec)
    3:  15 Delay After Ex (units = 0.01 sec)
    4:  0 mV Excitation

;Measure the HMP45C temperature in degrees Fahrenheit

3:  Volt (SE) (P1)
    1:  1 Reps
    2:  5 2500 mV Slow Range
    3:  3 SE Channel
    4:  1 Loc [ Tf       ]
    5:  0.18 Mult
    6:  -40 Offset

;Measure the HMP45C relative humidity in percent

4:  Volt (SE) (P1)
    1:  1 Reps
    2:  5 2500 mV Slow Range
    3:  4 SE Channel
    4:  2 Loc [ RH      ]
    5:  0.1 Mult
    6:  0.0 Offset
;Turn the HMP45C off

5: Do (P86)
   1: 51 Set Port 1 Low

;If the temperature is <80 degrees Fahrenheit,
;set the heat index value to the ambient temperature

6: If (X<=>F) (P89)
   1: 1 X Loc [ Tf ]
   2: 4 <
   3: 80 F
   4: 30 Then Do

7: Z = X (P31)
   1: 1 X Loc [ Tf ]
   2: 3 Z Loc [ HI ]

8: Else (P94)

;If the RH is <40%, set the heat index value
to the ambient temperature

9: If (X<=>F) (P89)
   1: 2 X Loc [ RH ]
   2: 4 <
   3: 40 F
   4: 30 Then Do

10: Z = X (P31)
    1: 1 X Loc [ Tf ]
    2: 3 Z Loc [ HI ]

11: Else (P94)

;Calculate the Heat Index (HI)
HI=-42.379+2.049*Tf+10.143*RH-0.22476*Tf*RH-
6.8378E-3*(Tf^2)-5.4817E-2*RH^2+1.2287E-3*(Tf^2)*RH+
8.5282E-4*RH^2*(Tf)-1.99E-6*Tf^2*RH^2

12: End (P95)
13: End (P95)
References and Resources

1. "Heat Index, Apparent Temperature";
   www.usatoday.com/weather/whumcalc.htm (updated 11/06/00)


3. "Humidity Vs. Heat Index";
   www.hutchison.org/weather/humidity/heat_index.html (updated 05/16/01)


5. "Your Free Heat Stress Monitor";
   www.ekginc.com/heatindex.htm


7. "Heat Index Calculator";
   www.web100.com/~sib/heatindex.html